

## Jordan-Normalform

Aufgaben: 8

```
> restart;  
with(LinearAlgebra):
```

## Ähnlichkeit von Matrizen

**MATH:** Es sei  $K$  ein Körper. Zwei Matrizen  $A, B \in K^{n \times n}$  heißen **ähnlich** oder **konjugiert**, unter  $GL(n, K)$ , falls ein  $g \in GL(n, K)$  existiert mit  $A = gBg^{-1}$ , d.h.  $A, B$  liegen in derselben Bahn unter der Operation

$$GL(n, K) \times K^{n \times n} \rightarrow K^{n \times n}: (g, C) \mapsto gCg^{-1}.$$

**DENKANSTOSS:** Formuliere obige Definition für Endomorphismen.

**BEISPIEL:** Basiswechsel

Es sei  $V := \mathbb{R}[x]_{\leq 3}$  der Vektorraum der Polynome vom Grad kleiner gleich 3 über den reellen Zahlen. Wir betrachten die formale Differentiation  $\Delta$  als Endomorphismus von  $V$ . Dann lässt sich  $\Delta$  bezüglich der beiden Basen  $B_1$  und  $B_2$  darstellen.

```
> B1 := [1, x, x^2, x^3];  
B1_Delta_B1 := Matrix(4, 4, (i, j) -> coeff(diff(B1[j], x), x, i-1));  
B1 := [1, x, x^2, x^3]
```

$$B1\_Delta\_B1 := \begin{bmatrix} 0 & 1 & 0 & 0 \\ 0 & 0 & 2 & 0 \\ 0 & 0 & 0 & 3 \\ 0 & 0 & 0 & 0 \end{bmatrix} \quad (1.1.1)$$

```
> B2 := map(n -> expand((x+n)^3), [$0..3]);  
B2 := [x^3, x^3 + 3x^2 + 3x + 1, x^3 + 6x^2 + 12x + 8, x^3 + 9x^2 + 27x + 27] \quad (1.1.2)
```

```
> B2_Delta_B2 := Matrix([[ -11/6, -1/3, 1/6, -1/3], [3, -1/2, -1, 3/2],  
[-3/2, 1, 1/2, -3], [1/3, -1/6, 1/3, 11/6]]);
```

$$B2\_Delta\_B2 := \begin{bmatrix} -\frac{11}{6} & -\frac{1}{3} & \frac{1}{6} & -\frac{1}{3} \\ 3 & -\frac{1}{2} & -1 & \frac{3}{2} \\ -\frac{3}{2} & 1 & \frac{1}{2} & -3 \\ \frac{1}{3} & -\frac{1}{6} & \frac{1}{3} & \frac{11}{6} \end{bmatrix} \quad (1.1.3)$$

Wir stellen nun die Basiswechselmatrix zwischen den beiden Basen auf und verifizieren, dass die Matrizen ähnlich sind.

> **T:=Matrix(4,4,(i,j)->coeff(B2[j],x,i-1));**

$$T:= \begin{bmatrix} 0 & 1 & 8 & 27 \\ 0 & 3 & 12 & 27 \\ 0 & 3 & 6 & 9 \\ 1 & 1 & 1 & 1 \end{bmatrix}$$

(1.1.4)

> **T^(-1).B1\_Delta\_B1.T=B2\_Delta\_B2;**

**T^(-1).B1\_Delta\_B1.T-B2\_Delta\_B2;**

**Equal(T^(-1).B1\_Delta\_B1.T,B2\_Delta\_B2);**

$$\begin{bmatrix} -\frac{11}{6} & -\frac{1}{3} & \frac{1}{6} & -\frac{1}{3} \\ 3 & -\frac{1}{2} & -1 & \frac{3}{2} \\ -\frac{3}{2} & 1 & \frac{1}{2} & -3 \\ \frac{1}{3} & -\frac{1}{6} & \frac{1}{3} & \frac{11}{6} \end{bmatrix} = \begin{bmatrix} -\frac{11}{6} & -\frac{1}{3} & \frac{1}{6} & -\frac{1}{3} \\ 3 & -\frac{1}{2} & -1 & \frac{3}{2} \\ -\frac{3}{2} & 1 & \frac{1}{2} & -3 \\ \frac{1}{3} & -\frac{1}{6} & \frac{1}{3} & \frac{11}{6} \end{bmatrix}$$

$$\begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

true

(1.1.5)

### ÜBUNG [01]:

Welche der folgenden Matrizen aus  $\mathbb{R}^4 \times 4$  sind ähnlich zueinander?

**A1,A2,A3,A4,A5:=Matrix([[1,2,2],[2,2,0],[2,0,0]]),Matrix([1,2,1],[2,1,1],[1,1,2]),Matrix([[2,0,2],[0,0,2],[2,2,1]]),Matrix([[2,1,1],[1,0,1],[1,1,0]]),Matrix([[1,1,2],[1,2,1],[2,1,1]]);**

$$A1, A2, A3, A4, A5:= \begin{bmatrix} 1 & 2 & 2 \\ 2 & 2 & 0 \\ 2 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 1 & 2 & 1 \\ 2 & 1 & 1 \\ 1 & 1 & 2 \end{bmatrix}, \begin{bmatrix} 2 & 0 & 2 \\ 0 & 0 & 2 \\ 2 & 2 & 1 \end{bmatrix}, \begin{bmatrix} 2 & 1 & 1 \\ 1 & 0 & 1 \\ 1 & 1 & 0 \end{bmatrix},$$

(1.1.6)

$$\begin{bmatrix} 1 & 1 & 2 \\ 1 & 2 & 1 \\ 2 & 1 & 1 \end{bmatrix}$$

Im Allgemeinen ist es allerdings schwer zu entscheiden, ob zwei Matrizen zueinander ähnlich sind. Bevor wir mit der Jordan-Normalform eine Möglichkeit dies zu tun kennen lernen, benötigen wir noch ein kleines Hilfsmittel.

**DENKANSTOSS:** Welche Eigenschaften von Matrizen kennst du bereits, die erfüllt sein müssen, damit diese ähnlich sein können?

## Hauptraumzerlegung

**MATH:** Es sei  $A \in K^{n \times n}$  eine Matrix und  $\mu_A(x) := \prod_{i=1}^l p_i^{m(i)}$  mit  $m(i), l \in \mathbb{N}$  die Zerlegung des Minimalpolynoms in normierte, irreduzible und paarweise verschiedene Polynome  $p_i$ . Dann existiert eine Zerlegung in die  $p_i$ -**Haupträume**  $\text{Kern}(p_i^{m(i)}(A))$ , welche alle  $A$ -invariant sind:

$$K^{n \times 1} = \bigoplus_{i=1}^l \text{Kern}(p_i^{m(i)}(A)).$$

Ferner gilt  $\text{Kern}(p_i^{m(i)}(A)) = \text{Bild}(q_i(A))$ , wobei  $q_i := \prod_{j \neq i} p_j^{m(j)}$  ist.

Die Projektion der Zerlegung sind durch  $\pi_i := (a_i q_i)(A)$  gegeben, wobei

$a_i \in K[x]$  mit  $1 = \sum_{i=1}^l a_i q_i$  z.B. durch den Euklidischen Algorithmus gegeben sind.

**BEISPIEL:** Wir betrachten die folgende Matrix  $A \in \mathbb{Q}^{7 \times 7}$ .

> **A:=CompanionMatrix(x^7-4\*x^6+9\*x^5-10\*x^4+4\*x^3+8\*x^2-12\*x+8);**

$$A := \begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 & -8 \\ 1 & 0 & 0 & 0 & 0 & 0 & 12 \\ 0 & 1 & 0 & 0 & 0 & 0 & -8 \\ 0 & 0 & 1 & 0 & 0 & 0 & -4 \\ 0 & 0 & 0 & 1 & 0 & 0 & 10 \\ 0 & 0 & 0 & 0 & 1 & 0 & -9 \\ 0 & 0 & 0 & 0 & 0 & 1 & 4 \end{bmatrix} \quad (1.2.1)$$

Zuerst bestimmen und faktorisieren wir das Minimalpolynom. Hierbei ist wichtig, dass wir später den Exponenten  $m(2) = 2$  nicht vergessen. Ferner bestimmen wir die Polynome  $q_i$ :

```

> factor(MinimalPolynomial(A,x));#störe dich nicht an der
Reihenfolge, Maple ordnet die Faktoren willkürlich
p1:=x+1;
p2:=x^2-2*x+2;
p3:=x^2-x+2;
q1:=p2^2*p3;
q2:=p1*p3;
q3:=p1*p2^2;

```

$$(x+1)(x^2-x+2)(x^2-2x+2)^2$$

$$p1:=x+1$$

$$p2:=x^2-2x+2$$

$$p3:=x^2-x+2$$

$$q1:=(x^2-2x+2)^2(x^2-x+2)$$

$$q2:=(x+1)(x^2-x+2)$$

$$q3:=(x+1)(x^2-2x+2)^2$$

(1.2.2)

Nun rechnen wir die Zerlegung der 1 mittels Euklidischem Algorithmus aus:

```

> gcdex(p1,q1,x,'b','a1'):
gcdex(p2^2,q2,x,'b','a2'):
gcdex(p3,q3,x,'b','a3'):

```

Und testen sie:

```

> a1*q1+a2*q2+a3*q3:
expand(%);

```

1

(1.2.3)

Damit sind wir nun in der Lage die Projektionen  $\pi_i$  zu bestimmen:

```

> #Dieses Programm setzt A in p ein
Eins:=proc(p::polynom, A::Matrix)
simplify(subs(x=A,expand(p)));
> end proc:
> pi1:=Eins(a1*q1,A);

```

(1.2.4)

$$\pi_1 := \begin{bmatrix} \frac{2}{25} & -\frac{2}{25} & \frac{2}{25} & -\frac{2}{25} & \frac{2}{25} & -\frac{2}{25} & \frac{2}{25} \\ -\frac{1}{5} & \frac{1}{5} & -\frac{1}{5} & \frac{1}{5} & -\frac{1}{5} & \frac{1}{5} & -\frac{1}{5} \\ \frac{7}{25} & -\frac{7}{25} & \frac{7}{25} & -\frac{7}{25} & \frac{7}{25} & -\frac{7}{25} & \frac{7}{25} \\ -\frac{6}{25} & \frac{6}{25} & -\frac{6}{25} & \frac{6}{25} & -\frac{6}{25} & \frac{6}{25} & -\frac{6}{25} \\ \frac{7}{50} & -\frac{7}{50} & \frac{7}{50} & -\frac{7}{50} & \frac{7}{50} & -\frac{7}{50} & \frac{7}{50} \\ -\frac{1}{20} & \frac{1}{20} & -\frac{1}{20} & \frac{1}{20} & -\frac{1}{20} & \frac{1}{20} & -\frac{1}{20} \\ \frac{1}{100} & -\frac{1}{100} & \frac{1}{100} & -\frac{1}{100} & \frac{1}{100} & -\frac{1}{100} & \frac{1}{100} \end{bmatrix} \quad (1.2.4)$$

Wir testen die Projektionseigenschaft:

**> pi1^2-pi1;**

$$\begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

(1.2.5)

**> pi2:=Eins(a2\*q2,A);**  
**pi3:=Eins(a3\*q3,A);**

$$\pi_2 := \begin{bmatrix} \frac{48}{25} & \frac{2}{25} & -\frac{52}{25} & -\frac{48}{25} & \frac{48}{25} & \frac{152}{25} & \frac{48}{25} \\ -\frac{4}{5} & \frac{9}{5} & \frac{16}{5} & \frac{4}{5} & -\frac{24}{5} & -\frac{36}{5} & \frac{16}{5} \\ -\frac{7}{25} & -\frac{18}{25} & -\frac{7}{25} & \frac{32}{25} & \frac{68}{25} & \frac{32}{25} & -\frac{132}{25} \\ \frac{31}{25} & -\frac{6}{25} & -\frac{44}{25} & -\frac{31}{25} & \frac{56}{25} & \frac{144}{25} & \frac{56}{25} \\ -\frac{89}{100} & \frac{57}{50} & \frac{59}{25} & \frac{16}{25} & -\frac{91}{25} & -\frac{134}{25} & \frac{84}{25} \\ \frac{3}{10} & -\frac{4}{5} & -\frac{6}{5} & \frac{1}{5} & \frac{14}{5} & \frac{16}{5} & -\frac{16}{5} \\ -\frac{1}{100} & \frac{13}{50} & \frac{6}{25} & -\frac{6}{25} & -\frac{19}{25} & -\frac{6}{25} & \frac{56}{25} \end{bmatrix}$$

$$\pi_3 := \begin{bmatrix} -1 & 0 & 2 & 2 & -2 & -6 & -2 \\ 1 & -1 & -3 & -1 & 5 & 7 & -3 \\ 0 & 1 & 1 & -1 & -3 & -1 & 5 \\ -1 & 0 & 2 & 2 & -2 & -6 & -2 \\ \frac{3}{4} & -1 & -\frac{5}{2} & -\frac{1}{2} & \frac{9}{2} & \frac{11}{2} & -\frac{7}{2} \\ -\frac{1}{4} & \frac{3}{4} & \frac{5}{4} & -\frac{1}{4} & -\frac{11}{4} & -\frac{9}{4} & \frac{13}{4} \\ 0 & -\frac{1}{4} & -\frac{1}{4} & \frac{1}{4} & \frac{3}{4} & \frac{1}{4} & -\frac{5}{4} \end{bmatrix}$$

(1.2.6)

Nun bestimmen wir Basen der Haupträume (Achtung bei  $p_2$ ):

> **BH1, BH2, BH3 := NullSpace(Eins(p1,A)), NullSpace(Eins(p2^2,A)), NullSpace(Eins(p3,A));**

$$BH1, BH2, BH3 := \left\{ \begin{bmatrix} 8 \\ -20 \\ 28 \\ -24 \\ 14 \\ -5 \\ 1 \end{bmatrix}, \begin{bmatrix} -4 \\ -4 \\ -1 \\ 0 \\ 0 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} -2 \\ -1 \\ 2 \\ 0 \\ 0 \\ 1 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 2 \\ 1 \\ 0 \\ 1 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 2 \\ 1 \\ 0 \\ 1 \\ 0 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 12 \\ -8 \\ -4 \\ 12 \\ -5 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} 4 \\ -4 \\ 0 \\ 4 \\ -3 \\ 1 \\ 0 \end{bmatrix} \right\} \quad (1.2.7)$$

Wir testen kurz, ob das Bild der Projektionen gleich den soeben bestimmten

Teilräumen ist. Dafür bestimmen wir zuerst das Bild von  $\pi_1$

> **ColumnSpace(pi1);**

$$\begin{bmatrix} 1 \\ -\frac{5}{2} \\ \frac{7}{2} \\ -3 \\ \frac{7}{4} \\ -\frac{5}{8} \\ \frac{1}{8} \end{bmatrix}$$

(1.2.8)

und vergleichen es mit der Basis  $BH_1$

> **Matrix([op(NullSpace(Eins(p1,A))),op(ColumnSpace(pi1))]);**  
**Rank(%);**

$$\begin{bmatrix} 8 & 1 \\ -20 & -\frac{5}{2} \\ 28 & \frac{7}{2} \\ -24 & -3 \\ 14 & \frac{7}{4} \\ -5 & -\frac{5}{8} \\ 1 & \frac{1}{8} \end{bmatrix}$$

1

(1.2.9)

Matrix hat Rang 1, also ist alles so wie erwartet. Der Vollständigkeit halber noch die restlichen Tests:

> **Matrix([op(NullSpace(Eins(p2^2,A))),op(ColumnSpace(pi2))]);**  
**Rank(%)=nops(NullSpace(Eins(p2^2,A)));**

$$\begin{bmatrix} 2 & -4 & -2 & 0 & 1 & 0 & 0 & 0 \\ 1 & -4 & -1 & 2 & 0 & 1 & 0 & 0 \\ 0 & -1 & 2 & 1 & 0 & 0 & 1 & 0 \\ 1 & 0 & 0 & 0 & 0 & 0 & 0 & 1 \\ 0 & 0 & 0 & 1 & -\frac{9}{16} & \frac{5}{8} & -\frac{1}{4} & \frac{1}{2} \\ 0 & 0 & 1 & 0 & \frac{1}{8} & -\frac{1}{4} & \frac{1}{2} & 0 \\ 0 & 1 & 0 & 0 & -\frac{5}{16} & \frac{1}{8} & -\frac{1}{4} & \frac{1}{2} \end{bmatrix}$$

$$4 = 4$$

(1.2.10)

> **Matrix([op(NullSpace(Eins(p3,A))),op(ColumnSpace(pi3))]);**  
**Rank(%)=nops(NullSpace(Eins(p3,A)));**

$$\begin{bmatrix} 12 & 4 & 1 & 0 \\ -8 & -4 & 0 & 1 \\ -4 & 0 & -1 & -1 \\ 12 & 4 & 1 & 0 \\ -5 & -3 & \frac{1}{4} & 1 \\ 0 & 1 & -\frac{1}{2} & -\frac{3}{4} \\ 1 & 0 & \frac{1}{4} & \frac{1}{4} \end{bmatrix}$$

$$2 = 2$$

(1.2.11)

Nun können wir noch den Basiswechsel bei der Matrix  $A$  durchführen und sollten eine Blockdiagonalmatrix erhalten.

> **T:=Matrix([op(BH1),op(BH2),op(BH3)]);**  
**T^(-1).A.T;**

$$T := \begin{bmatrix} 8 & -4 & -2 & 0 & 2 & 12 & 4 \\ -20 & -4 & -1 & 2 & 1 & -8 & -4 \\ 28 & -1 & 2 & 1 & 0 & -4 & 0 \\ -24 & 0 & 0 & 0 & 1 & 12 & 4 \\ 14 & 0 & 0 & 1 & 0 & -5 & -3 \\ -5 & 0 & 1 & 0 & 0 & 0 & 1 \\ 1 & 1 & 0 & 0 & 0 & 1 & 0 \end{bmatrix}$$



$$\begin{bmatrix} -1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 4 & 1 & 0 & 0 & 0 & 0 \\ 0 & -9 & 0 & 1 & 0 & 0 & 0 \\ 0 & 10 & 0 & 0 & 1 & 0 & 0 \\ 0 & -5 & 2 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 4 & 1 \\ 0 & 0 & 0 & 0 & 0 & -14 & -3 \end{bmatrix}$$

(1.2.12)

**DENKANSTOSS:** Erkläre das folgende "Wunder".

> T.DiagonalMatrix([1\$1,0\$4,0\$2]).T^(-1)-pi1;  
 T.DiagonalMatrix([0\$1,1\$4,0\$2]).T^(-1)-pi2;  
 T.DiagonalMatrix([0\$1,0\$4,1\$2]).T^(-1)-pi3;

$$\begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

$$\begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

$$\begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

(1.2.13)

## ÜBUNG [02]:

Gib eine Matrix  $T$  an, welche die Matrix  $B$  über dem Körper  $\mathbb{Q}$  der rationalen Zahlen mit Blöcken möglichst kleinen Grades auf Blockdiagonalgestalt konjugiert. (Hinweis: Der Befehl **factor** kann benutzt werden.)

> **B:=CompanionMatrix(x^10-4\*x^9-4\*x^8+12\*x^7+12\*x^6+32\*x^5-48\*x^4-48\*x^3-64\*x^2-64\*x+256);**

$$B := \begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -256 \\ 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 64 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 64 \\ 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 48 \\ 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 48 \\ 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & -32 \\ 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & -12 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & -12 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 4 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 4 \end{bmatrix}$$

[

## ▼ Zyklische Vektoren

**MATH:** Es sei  $\alpha \in \text{End}(V)$  ein Endomorphismus des  $K$ -Vektorraumes  $V$ . Ein Vektor  $v \in V$  heißt **zyklischer Vektor** von  $V$  (bzgl.  $\alpha$ ), falls  $\langle v \rangle_\alpha := K[\alpha](v) = V$  gilt. Falls ein solcher existiert, so heißt  $V$  **zyklischer Vektorraum** (bzgl.  $\alpha$ ).

**BEISPIEL:**

> **A:=Matrix([[1,2,1,-1],[-1,-1,1,0],[0,-1,-6,3],[0,-2,-12,6]])**  
;

$$A := \begin{bmatrix} 1 & 2 & 1 & -1 \\ -1 & -1 & 1 & 0 \\ 0 & -1 & -6 & 3 \\ 0 & -2 & -12 & 6 \end{bmatrix}$$

(1.3.1)

Wir behaupten der folgende Vektor sei ein zyklischer:

> **v:=Vector([2,-2,-1,-1]);**

$$v := \begin{bmatrix} 2 \\ -2 \\ -1 \\ -1 \end{bmatrix}$$

(1.3.2)

Dies testen wir:

```
> map(i->A^i.v,[$0..3]);
Matrix(%);
Rank(%);
```

$$\left[ \begin{bmatrix} 2 \\ -2 \\ -1 \\ -1 \end{bmatrix}, \begin{bmatrix} -2 \\ -1 \\ 5 \\ 10 \end{bmatrix}, \begin{bmatrix} -9 \\ 8 \\ 1 \\ 2 \end{bmatrix}, \begin{bmatrix} 6 \\ 2 \\ -8 \\ -16 \end{bmatrix} \right]$$

$$\begin{bmatrix} 2 & -2 & -9 & 6 \\ -2 & -1 & 8 & 2 \\ -1 & 5 & 1 & -8 \\ -1 & 10 & 2 & -16 \end{bmatrix}$$

4

(1.3.3)

**DENKANSTOSS:** Warum genügt es die Potenzen von  $A$  bis hin zu  $A^3$  zu nehmen?

Leider sind nicht alle Vektoren zyklisch, so ist zum Beispiel nur die Hälfte der Standardbasisvektoren in diesem Fall zyklisch:

```
> B:=[Vector([1,0,0,0]),Vector([0,1,0,0]),Vector([0,0,1,0]),
Vector([0,0,0,1])];
for e in B do
Matrix(map(i->A^i.e,[$0..3]));
Rank(%);
od;
```

$$B := \left[ \begin{bmatrix} 1 \\ 0 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 1 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 1 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 0 \\ 1 \end{bmatrix} \right]$$

$$\begin{bmatrix} 1 & 1 & -1 & -2 \\ 0 & -1 & 0 & 2 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 2 & 0 \end{bmatrix}$$

3

$$\begin{bmatrix} 0 & 2 & 1 & -4 \\ 1 & -1 & -2 & 2 \\ 0 & -1 & 1 & 2 \\ 0 & -2 & 2 & 4 \end{bmatrix}$$

3

$$\begin{bmatrix} 0 & 1 & 9 & -6 \\ 0 & 1 & -8 & -2 \\ 1 & -6 & -1 & 8 \\ 0 & -12 & -2 & 16 \end{bmatrix}$$

4

$$\begin{bmatrix} 0 & -1 & -4 & 4 \\ 0 & 0 & 4 & 0 \\ 0 & 3 & 0 & -4 \\ 1 & 6 & 0 & -8 \end{bmatrix}$$

4

(1.3.4)

### ÜBUNG [03]:

Gib eine Basis an, sodass der Endomorphismus gegeben durch  $A$  durch die Begleitmatrix des Minimalpolynoms von  $A$  dargestellt wird.

### ÜBUNG [04]:

Eine Matrix  $B \in K^{n \times n}$  heißt **nilpotent**, falls ein  $k \in \mathbb{N}$  existiert mit  $B^k = 0$ .

1.) Zeige, dass  $k \leq n$  gewählt werden kann.

2.) Gib für  $n = 5$  eine nilpotente Matrix  $B$  und einen Vektor  $v$  an, so dass  $K^{5 \times 1}$  bezüglich  $B$  zyklisch von  $v$  erzeugt wird.

**WICHTIG:** Durch die Hauptraumzerlegung motiviert nehmen wir im Folgenden stets an, dass das Minimalpolynom nur einen einzigen irreduziblen Faktor  $p$  besitzt.

**DENKANSTOSS:** Warum genau ist die Annahme gerechtfertigt?

Nun haben wir ein leichtes Kriterium, um zu entscheiden, ob  $V$  zyklisch ist:

**MATH:** Sei  $\alpha \in \text{End}(V)$  mit  $\mu_\alpha = p^m$  für  $p \in K[x]$  irreduzibel. Definiere

$d := \text{Grad}(p)$  und  $v := p(\alpha)$ . Dann sind äquivalent:

- 1)  $V$  ist zyklisch bezüglich  $\alpha$ .
- 2)  $\text{Kern}(v)$  hat Dimension  $d$  über  $K$ .
- 3)  $v$  hat Rang  $\dim(V) - d$ .
- 4)  $\mu_\alpha = \chi_\alpha$
- 5) Sämtliche  $\alpha$ -invarianten Teilräume von  $V$  sind gegeben durch  $\langle v^j(V) \rangle_\alpha = \text{Bild}(v^i)$  für  $i = 0, 1, \dots, m$ .

Insbesondere die  $\alpha$ -invarianten Teilräume linear durch Inklusion geordnet.

**BEISPIEL:**

```
> A:=Matrix([[-11,0,23,22],[-9,-1,21,21],[-8,2,13,12],[4,-2,-5,-5]]);  
MinimalPolynomial(A,x);  
CharacteristicPolynomial(A,x);  
factor(%);  
nu:=Eins(x+1,A);
```

$$A := \begin{bmatrix} -11 & 0 & 23 & 22 \\ -9 & -1 & 21 & 21 \\ -8 & 2 & 13 & 12 \\ 4 & -2 & -5 & -5 \end{bmatrix}$$

$$x^4 + 4x^3 + 6x^2 + 4x + 1$$

$$x^4 + 4x^3 + 6x^2 + 4x + 1$$

$$(x+1)^4$$

$$v := \begin{bmatrix} -10 & 0 & 23 & 22 \\ -9 & 0 & 21 & 21 \\ -8 & 2 & 14 & 12 \\ 4 & -2 & -5 & -4 \end{bmatrix}$$

(1.3.5)

Wir sind also im zyklischen Fall.

```
> nu:=Eins(x+1,A);  
NullSpace(nu);
```

$$v := \begin{bmatrix} -10 & 0 & 23 & 22 \\ -9 & 0 & 21 & 21 \\ -8 & 2 & 14 & 12 \\ 4 & -2 & -5 & -4 \end{bmatrix}$$

$$\begin{bmatrix} -7 \\ -6 \\ -4 \\ 1 \end{bmatrix}$$

(1.3.6)

Wir haben Glück und der erste Standardbasisvektor ist ein zyklischer Erzeuger.

```
> v:=Vector([1,0,0,0]);
   Matrix(map(i->A^i.v,[0..3])):Rank(%);
```

$$v:=\begin{bmatrix} 1 \\ 0 \\ 0 \\ 0 \end{bmatrix}$$

4

(1.3.7)

Wir bilden die zyklische Basis und stellen fest, dass der Wechsel zu dieser Basis  $A$  in eine besonders hübsche Form bringt.

```
> B:=map(i->nu^i.v,[0..3]);
   T:=Matrix(B);
   T^(-1).A.T;
```

$$B:=\left[ \begin{bmatrix} 1 \\ 0 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} -10 \\ -9 \\ -8 \\ 4 \end{bmatrix}, \begin{bmatrix} 4 \\ 6 \\ -2 \\ 2 \end{bmatrix}, \begin{bmatrix} -42 \\ -36 \\ -24 \\ 6 \end{bmatrix} \right]$$

$$T:=\begin{bmatrix} 1 & -10 & 4 & -42 \\ 0 & -9 & 6 & -36 \\ 0 & -8 & -2 & -24 \\ 0 & 4 & 2 & 6 \end{bmatrix}$$

$$\begin{bmatrix} -1 & 0 & 0 & 0 \\ 1 & -1 & 0 & 0 \\ 0 & 1 & -1 & 0 \\ 0 & 0 & 1 & -1 \end{bmatrix}$$

(1.3.8)

Dies war aber kein Zufall.

**MATH:** Sei  $\alpha \in \text{End}(V)$  mit  $\mu_\alpha = p^m$  für  $p \in K[x]$  irreduzibel. Definiere  $d := \text{Grad}(p)$  und  $v := p(\alpha)$  sowie  $n := m \cdot d = \text{Dim}(V)$ . Ist  $v$  ein zyklischer Erzeuger von  $V$ , so ist  $B \in V^n$  definiert durch

$$B_{r+d \cdot i} := \alpha^{r-1} v^i(v) \quad \text{für } 1 \leq r \leq d \text{ und } 0 \leq i \leq \frac{n}{d}$$

eine Basis von  $V$  und es gilt  ${}^B \alpha^B = J(p^m)$ , wobei  $J(p^m)$  der **verallgemeinerte Jordanblock** von  $p^m$  ist, definiert durch

$$J(p^m) := \begin{bmatrix} M_p & 0 & 0 & \dots & 0 & 0 \\ N_d & M_p & 0 & \dots & 0 & 0 \\ 0 & N_d & M_p & \dots & 0 & 0 \\ 0 & 0 & N_d & \dots & 0 & 0 \\ \vdots & \vdots & \vdots & \vdots & \vdots & \vdots \\ 0 & 0 & 0 & \dots & N_d & M_p \end{bmatrix} \in K^{md \times md}, \text{ wobei } 0 \in K^{d \times d} \text{ die } d \times d$$

Nullmatrix und  $N_d := \{1, \dots, d\} \times \{1, \dots, d\} \rightarrow K: (i, j) \mapsto \begin{cases} 1, & \text{falls } (i, j) = (1, d) \\ 0, & \text{sonst} \end{cases}$ .

**MATH:** Das folgende Programm erwartet ein irreduzibles Polynom  $p$  und eine natürliche Zahl  $m$  und gibt den zugehörigen (verallgemeinerten) **Jordan-Block**  $J(p^m)$  aus:

```
> JBlock:=proc(p, x::name, m::posint,
characteristic::nonnegint:=0)
local M,d,i,irr:
if not type(p, polynom(anything, x)) or degree(p,x) < 1
then
error sprintf("p ist kein Polynom in %a.", x):
fi:
if characteristic > 0 then
if not isprime(characteristic) then
error "Charakteristik muss eine Primzahl sein.":
fi:
irr := Irreduc(p) mod characteristic:
else
irr := irreduc(p):
fi:
if not irr then
error "p ist nicht irreduzibel.":
fi:
d := degree(p, x):
M := LinearAlgebra[DiagonalMatrix]([(LinearAlgebra
[CompanionMatrix](p, x))$m]):
for i from 1 to m-1 do
M[d*i+1,d*i]:=1:
od:
M:
end:
```

> **JBlock(x+1, x, 4);**

$$\begin{bmatrix} -1 & 0 & 0 & 0 \\ 1 & -1 & 0 & 0 \\ 0 & 1 & -1 & 0 \\ 0 & 0 & 1 & -1 \end{bmatrix}$$

(1.3.9)

Wir behandeln hier zusätzlich **verallgemeinerte Jordan-Blöcke**. So lässt sich ein Jordan-Block für eine Potenz eines irreduziblen Polynoms angeben, welches nicht Grad 1 hat:

> **JBlock(x^2+x+1, x, 3);**

$$\begin{bmatrix} 0 & -1 & 0 & 0 & 0 & 0 \\ 1 & -1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & -1 & 0 & 0 \\ 0 & 0 & 1 & -1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & -1 \\ 0 & 0 & 0 & 0 & 1 & -1 \end{bmatrix}$$

(1.3.10)

Letzteres ist ein verallgemeinerter Jordan-Block für  $K = \mathbb{Q}$  oder  $K = \mathbb{R}$ . Der verallgemeinerte Jordan-Block ist aber weiterhin nur für irreduzible Polynome definiert. Über  $\mathbb{C}$  gibt es keinen Jordan-Block zu  $x^2 + x + 1$ , wohl aber zu seinen irreduziblen Faktoren:

> **solve(x^2+x+1, x);**

$$-\frac{1}{2} + \frac{1}{2}i\sqrt{3}, -\frac{1}{2} - \frac{1}{2}i\sqrt{3}$$

(1.3.11)

> **JBlock(x-(-1/2+(1/2\*I)\*sqrt(3)), x, 3), JBlock(x-(-1/2-(1/2\*I)\*sqrt(3)), x, 3);**

$$\begin{bmatrix} -\frac{1}{2} + \frac{1}{2}i\sqrt{3} & 0 & 0 \\ 1 & -\frac{1}{2} + \frac{1}{2}i\sqrt{3} & 0 \\ 0 & 1 & -\frac{1}{2} + \frac{1}{2}i\sqrt{3} \end{bmatrix},$$

(1.3.12)

$$\begin{bmatrix} -\frac{1}{2} - \frac{1}{2}i\sqrt{3} & 0 & 0 \\ 1 & -\frac{1}{2} - \frac{1}{2}i\sqrt{3} & 0 \\ 0 & 1 & -\frac{1}{2} - \frac{1}{2}i\sqrt{3} \end{bmatrix}$$

Das bedeutet, dass wir über  $\mathbb{C}$  eine Zerlegung in zwei Haupträume haben,



während über  $\mathbb{Q}$  oder  $\mathbb{R}$  nur ein Hauptraum vorliegt. Weiteres Beispiel:  $x^2 + 1$  ist irreduzibel über  $\mathbb{Q}$ , nicht aber über  $\mathbb{F}_2$ :

```
> JBlock(x^2+1, x, 2);
```

$$\begin{bmatrix} 0 & -1 & 0 & 0 \\ 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & -1 \\ 0 & 0 & 1 & 0 \end{bmatrix}$$

(1.3.13)

```
> JBlock(x^2+1, x, 2, 2);
```

Error. (in JBlock) p ist nicht irreduzibel.

Der verallgemeinerte Jordanblock ermöglicht es uns, eine (verallgemeinerte) Jordan-Normalform zu definieren, die nicht erfordert, dass  $K$  algebraisch abgeschlossen ist

**MAPLE** hat eine etwas andere Auffassung davon, was eine solche Jordan-Block-Matrix ist und liefert auch nur etwas, wenn  $p$  vom Grad 1, ist:

```
> JordanBlockMatrix([[ -1, 4 ]]);
```

$$\begin{bmatrix} -1 & 1 & 0 & 0 \\ 0 & -1 & 1 & 0 \\ 0 & 0 & -1 & 1 \\ 0 & 0 & 0 & -1 \end{bmatrix}$$

(1.3.14)

Wir hätten diese Matrix transponiert, weil wir eine andere Konvention benutzen, die durch den nilpotenten Fall von oben motiviert ist. (Überprüfe die Konvention in deiner Vorlesung.)

### ÜBUNG [05]:

Man gebe die Matrix an, die die Begleitmatrix von  $(x-1)^5$  auf ihren Jordanblock transformiert.

```
> B:=CompanionMatrix((x-1)^5,x);
```

$$B:=\begin{bmatrix} 0 & 0 & 0 & 0 & 1 \\ 1 & 0 & 0 & 0 & -5 \\ 0 & 1 & 0 & 0 & 10 \\ 0 & 0 & 1 & 0 & -10 \\ 0 & 0 & 0 & 1 & 5 \end{bmatrix}$$

(1.3.15)

## Zerlegung eines Hauptraumes in zyklische Räume

**MATH:** Ist  $V$  nicht mehr zyklisch bezüglich  $A$ , aber immer noch mit Minimalpolynom  $p^m$  und charakteristischem Polynom  $p^c$ , wo  $p \in K[x]$  normiert und irreduzibel ist ( $m < c$ ), so kann man  $V$  in eine direkte Summe bezüglich  $A$  zyklischer Teilräume zerlegen. Z. B.

$$V := \langle v_1, Av_1, A^2v_1, \dots \rangle \oplus \langle v_2, Av_2, A^2v_2, \dots \rangle \oplus \dots = \langle v_1 \rangle_A \oplus \langle v_2 \rangle_A \oplus \dots$$

Dass  $V$  Summe solcher zyklischer Teilräume ist, ist wegen der endlichen Dimension klar (**DENKANSTOSS:** Warum?). Unsere Aufgabe besteht also darin, von der Summe zur direkten Summe überzugehen.

**MATH:** Sei  $\alpha \in \text{End}(V)$  mit  $\mu_\alpha = p^m$  für ein irreduzibles  $p \in K[x]$ . und

$v := p(\alpha)$ . Für  $w \in V$  ist die  $\alpha$ -Länge von  $w$  definiert als  $\lambda(w) := \min\{i \in \mathbb{N} : v^i(w) = 0\}$ .

**BEISPIEL:**

>  $p := x^2 + x + 1$ ;

$$p := x^2 + x + 1 \quad (1.4.1)$$

>  $A := \text{DiagonalMatrix}([\text{CompanionMatrix}(p^3, x), \text{CompanionMatrix}(p, x)])$ ;

>  $T := \text{SubMatrix}(\text{Transpose}(A), 1..8, [8, 7, 6, 5, 4, 3, 2, 1])^2$ ;

>  $A := T \cdot A \cdot T^{-1}$ ;

$$A := \begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 1 & 1 \\ -6 & -22 & -39 & -49 & -43 & -3 & -109 & -96 \\ 2 & 6 & 7 & 6 & 3 & 1 & 6 & 6 \\ 0 & 0 & 1 & 0 & 0 & 0 & 7 & 7 \\ 0 & 0 & 0 & 1 & 0 & 0 & 6 & 6 \\ 23 & 90 & 179 & 249 & 237 & 11 & 553 & 475 \\ 0 & 1 & 3 & 6 & 7 & 0 & -3 & -6 \\ 0 & -1 & -3 & -6 & -8 & 0 & 0 & 3 \end{bmatrix} \quad (1.4.2)$$

Wir wissen dass  $p^3$  das Minimalpolynom von  $A$  ist:

>  $\text{Eins}(p^3, A), \text{Eins}(p^2, A)$ ;

(1.4.3)

$$\begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix},$$

(1.4.3)

$$\begin{bmatrix} 1 & 0 & -1 & 1 & 0 & 0 & 1 & 1 \\ -116 & -264 & -236 & -292 & -176 & -44 & -72 & -72 \\ 7 & 12 & 9 & 15 & 8 & 2 & 5 & 5 \\ 8 & 18 & 16 & 20 & 12 & 3 & 5 & 5 \\ 5 & 12 & 11 & 13 & 8 & 2 & 3 & 3 \\ 586 & 1362 & 1230 & 1494 & 908 & 227 & 359 & 359 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ -2 & -6 & -6 & -6 & -4 & -1 & -1 & -1 \end{bmatrix}$$

Wir haben wieder die nilpotente Matrix

**> nu:=Eins(p,A):  
nu,nu^3;**

$$\begin{bmatrix} 1 & 0 & 0 & 0 & -1 & 0 & -2 & -2 \\ -21 & -54 & -79 & -73 & 80 & -9 & 116 & 116 \\ 3 & 6 & 8 & 6 & -3 & 1 & -9 & -9 \\ 2 & 6 & 8 & 7 & -4 & 1 & -8 & -8 \\ 0 & 0 & 1 & 1 & -5 & 0 & -5 & -5 \\ 94 & 252 & 374 & 357 & -418 & 42 & -571 & -571 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & -1 & 2 & 0 & 1 & 1 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

(1.4.4)

**> E1:=Column(IdentityMatrix(8),1):**

**> T1:=Matrix(map(i->(A^i).E1,[\$0..5])); Rank(T1);  
Matrix(map(i->(A^i).E1,[\$0..6])); Rank(%);**

6

6

(1.4.5)

Also hat der von  $E_1$  erzeugte  $A$ -zyklische Teilraum  $\langle E_1 \rangle_A$  die Dimension 6.

Ebenso der von

$E_2$  erzeugt:

```
> E2:=Column(IdentityMatrix(8),2);  
> T2:=Matrix(map(i->(A^i).E2,[$0..5])): Rank(T2);  
Matrix(map(i->(A^i).E2,[$0..6])): Rank(%);  
6  
6
```

(1.4.6)

```
> Rank(<T1|T2>);  
8
```

(1.4.7)

Also ist klar: Die Summe der von  $E_1$  und von  $E_2$  erzeugten zyklischen Teilräume von  $V$  ist ganz  $V$ . Auch ist klar, dass der Schnitt 4-dimensional ist:

- 1.)  $\langle E_1 \rangle_A + \langle E_2 \rangle_A = V$
- 2.)  $\dim(\langle E_1 \rangle_A \cap \langle E_2 \rangle_A) = 4.$

Also nach unseren Einsichten über zyklische Räume für  $A$  gilt, da insbesondere der Schnitt  $A$ -invarianten Räumen wieder  $A$ -invariant ist und wir damit seine Form kennen:

$$\langle E_1 \rangle_A \cap \langle E_2 \rangle_A = v(\langle E_1 \rangle_A)$$

Wir versuchen,  $E_2$  durch einen anderen Vektor zu ersetzen, wo der Schnitt kleiner, aber das Gesamterzeugnis immer noch ganz  $V$  ist.

Da  $v^3 E_2 = 0$  aber

```
> (nu^2).E2;  
  
      0  
     -264  
      12  
      18  
      12  
     1362  
      0  
      -6
```

(1.4.8)

noch ungleich Null, versuchen wir  $E_2$  so zu modifizieren, dass wir Null bekommen. Also versuchen wir  $v^2 E_2$  linear zu kombinieren aus  $v^2(E_1)$  und  $v^2(AE_1)$ , die eine  $K$ -Basis des kleinsten, nicht-trivialen  $A$ -invarianten Teilraumes von  $\langle E_1 \rangle_A$  sind:

```
> Test1:=<(nu^2).E1|(nu^2).A.E1|(nu^2).E2>;
```

$$Test1 := \begin{bmatrix} 1 & -2 & 0 \\ -116 & 100 & -264 \\ 7 & -8 & 12 \\ 8 & -7 & 18 \\ 5 & -4 & 12 \\ 586 & -491 & 1362 \\ 0 & 0 & 0 \\ -2 & 1 & -6 \end{bmatrix} \quad (1.4.9)$$

> nn:=op(NullSpace(Test1));

$$nn := \begin{bmatrix} -4 \\ -2 \\ 1 \end{bmatrix} \quad (1.4.10)$$

Wir ersetzen  $E_2$  durch

> E2n:=<E1|A.E1|E2>.nn;

$$E2n := \begin{bmatrix} -4 \\ 13 \\ -4 \\ 0 \\ 0 \\ -46 \\ 0 \\ 0 \end{bmatrix} \quad (1.4.11)$$

**DENKANSTOSS:** Warum ist immer noch  $\langle E_1, E_{2n} \rangle_A = V$ ?

> (nu^2).E2n,nu.E2n;

(1.4.12)

$$\begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} -4 \\ 112 \\ -12 \\ -8 \\ -4 \\ -528 \\ 0 \\ 0 \end{bmatrix}$$

(1.4.12)

Wir haben also einen Fortschritt:  $E_{2n}$  hat Länge 2 statt 3, d. h. es wird nach 2-maligem Anwenden von  $v$  erstmalig 0.

**DENKANSTOSS:**  $\dim(\langle E_1 \rangle_A \cap \langle E_{2n} \rangle_A) = 2.$

Wir wollen aber aus Dimensionsgründen eines mit Länge 1. Also wiederholen wir den Erfolg:

**> Test2:=<(nu^2).E1|(nu^2).A.E1|(nu).E2n>;**

$$Test2 := \begin{bmatrix} 1 & -2 & -4 \\ -116 & 100 & 112 \\ 7 & -8 & -12 \\ 8 & -7 & -8 \\ 5 & -4 & -4 \\ 586 & -491 & -528 \\ 0 & 0 & 0 \\ -2 & 1 & 0 \end{bmatrix}$$

(1.4.13)

**> nn:=3\*op(NullSpace(Test2));**

$$nn := \begin{bmatrix} -4 \\ -8 \\ 3 \end{bmatrix}$$

(1.4.14)

Wir ersetzen  $E_{2n}$  durch (Hier gut aufpassen!):

**> E2nn:=<nu.E1|nu.A.E1|E2n>.nn;**

(1.4.15)

$$E_{2nn} := \begin{bmatrix} -16 \\ 451 \\ -48 \\ -32 \\ -16 \\ -2130 \\ 0 \\ 0 \end{bmatrix} \quad (1.4.15)$$

> nu.E2nn;

$$\begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix} \quad (1.4.16)$$

Jetzt hat  $E_{2nn}$  Länge 1. Immer noch erzeugt es mit  $E_1$  zusammen  $V$ , wenn man die Bilder unter  $A$  hinzunimmt. Also: Unsere Zauberbasis sind die Spalten von:

> T:=Matrix([E1,A.E1,nu.E1,A.nu.E1,nu.nu.E1,A.nu.nu.E1,E2nn,A.E2nn]);

$$T := \begin{bmatrix} 1 & 0 & 1 & 0 & 1 & -2 & -16 & 0 \\ 0 & -6 & -21 & -41 & -116 & 100 & 451 & 692 \\ 0 & 2 & 3 & 3 & 7 & -8 & -48 & -32 \\ 0 & 0 & 2 & 3 & 8 & -7 & -32 & -48 \\ 0 & 0 & 0 & 2 & 5 & -4 & -16 & -32 \\ 0 & 23 & 94 & 202 & 586 & -491 & -2130 & -3560 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 3 \\ 0 & 0 & 0 & 0 & -2 & 1 & 0 & 13 \end{bmatrix} \quad (1.4.17)$$

> (T^(-1)).A.T,DiagonalMatrix([JBlock(p,x,3),JBlock(p,x,1)]);

$$\begin{pmatrix} 0 & -1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 1 & -1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & -1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & -1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & -1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & -1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1 \end{pmatrix}, \begin{pmatrix} 0 & -1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 1 & -1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & -1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & -1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & -1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & -1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1 \end{pmatrix} \quad (1.4.18)$$

**DENKANSTOSS:** Warum haben wir mit  $(\nu^2).E1$  und  $(\nu^2).A.E1$  statt mit  $\nu.E1$  und  $\nu.A.E1$  gerechnet? Zeichne dir am besten ein Diagramm der Inklusionen der invarianten Räume.

### ÜBUNG [06]:

Zerlege den Vektorraum  $\mathbb{Q}^4 \times 1$  in eine direkte Summe von bezüglich  $B$  zyklischen Teilräumen, mit

**> B := Matrix([[ -2050, -2965, -3211, -2401 ], [ 1152, 1344, 1152, 576 ], [ 0, 0, 0, 0 ], [ 4, 298, 598, 706 ]]);**

$$B := \begin{pmatrix} -2050 & -2965 & -3211 & -2401 \\ 1152 & 1344 & 1152 & 576 \\ 0 & 0 & 0 & 0 \\ 4 & 298 & 598 & 706 \end{pmatrix} \quad (1.4.19)$$

und benutze dies um (wie oben) die Jordan-Normalform herzustellen.

(Hinweis: Wende das eben vorgeführte Verfahren an. Warum ist es in diesem Fall einfacher? Warum muss man dennoch besonders aufpassen?)

Wir hatten nicht immer nur Glück, sondern eine solche Zerlegung lässt sich stets finden:

**MATH:** Sei  $\alpha \in \text{End}(V)$  mit  $\mu_\alpha = p^m$  für ein irreduzibles, normiertes  $p \in K[x]$ . Ist  $v \in V$  mit  $\mu_{\alpha, v} = p^m$  und  $w \in V$  beliebig, so existiert ein  $w_1 \in V$  mit:  $\langle v \rangle_\alpha + \langle w \rangle_\alpha = \langle v \rangle_\alpha \oplus \langle w_1 \rangle_\alpha$

Bew.: Sei  $\mu_{\alpha, w} = p^r$ . Der Schnitt von  $\langle v \rangle_\alpha$  und  $\langle w \rangle_\alpha$  ist genau dann nicht leer, wenn  $v^{r-1}(\langle w \rangle_\alpha)$  in  $\langle v \rangle_\alpha$  enthalten ist. Finde nun durch Lösen eines LGS ein  $q \in K[x]$  mit kleinerem Grad als  $p$  mit:  $q(\alpha)v^{m-1}(v) = v^{r-1}(w)$ . Nun ersetze  $w$



(beachte  $r \leq m$ ) durch  $w' := w - q(\alpha)v^{m-r}(v)$ . Nun gilt  $\mu_{\alpha, w'} = p^s$  mit  $s < r$ . Ferner ist  $\langle v \rangle_{\alpha} + \langle w \rangle_{\alpha} = \langle v \rangle_{\alpha} + \langle w' \rangle_{\alpha}$ . Nach endlich vielen Schritten liegt eine direkte Summe vor.

**MATH:** Der Beweis des vorherigen Satzes liefert uns den Algorithmus, welchen wir oben bereits (unbewusst) verwendet haben.

Es folgt sofort:

**MATH:** Sei  $\alpha \in \text{End}(V)$  mit  $\mu_{\alpha} = p^m$  für ein irreduzibles  $p \in K[x]$ . Dann existiert

ein  $k \in \mathbb{N}$  und Vektoren  $v_1, \dots, v_k \in V$  mit:  $V = \bigoplus_{i=1}^k \langle v_i \rangle_{\alpha}$

## Die Partition eines Endomorphismus

**MATH:** Eine **Partition** der natürlichen Zahl  $c$  ist ein  $k$ -Tupel  $a = (a_1, \dots, a_k) \in \mathbb{N}^k$  mit der Eigenschaft  $a_1 \geq \dots \geq a_k$  und  $a_1 + \dots + a_k = c$ . Ist  $a = (a_1, \dots, a_k) \in \mathbb{N}^k$  eine Partition von  $c \in \mathbb{N}$ , so ist die **konjugierte Partition**  $a'$  von  $a$  definiert durch  $a'_i := |\{j : a_j \geq i\}|$ . Sie ist ebenfalls eine Partition von  $c$ .

**BEISPIEL:**

Wir nehmen die Partition  $(5, 3, 2)$  von 10 und visualisieren die durch

```

x x x x x
x x x
x x

```

Dieses Schema wird an der Diagonalen gespiegelt (transponiert):

```

x x x
x x x
x x
x
x

```

Nun erhalten wir die konjugierte Partition  $(3, 3, 2, 1, 1)$ .

Wir stellen fest, dass ein Zusammenhang zwischen Partitionen und Zerlegungen besteht:

**MATH:** Sei  $\alpha \in \text{End}(V)$  mit  $\mu_{\alpha} = p^m$  und  $\chi_{\alpha} = p^c$  für ein irreduzibles  $p \in K[x]$ . Eine

Zerlegung  $V = \bigoplus_{i=1}^k \langle v_i \rangle_{\alpha}$  mit  $\lambda(v_1) \geq \dots \geq \lambda(v_k) > 0$  heißt der Länge nach

geordnet. Ihr wird die Partition  $(\lambda(v_1), \dots, \lambda(v_k))$  von  $c$  zugeordnet. Je zwei der Länge nach geordneten Zerlegungen von  $V$  ist dieselbe Partition von  $c$  zugeordnet. Diese heißt daher die zu  $\alpha$  **gehörige Partition**.

**BEISPIEL:** Wir betrachten erneut die Matrix  $A$  aus dem vorherigem Abschnitt.

Dort hatten wir E1 und E2nn als Vektoren einer Zerlegung gefunden.

> A;

$$\begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 1 & 1 \\ -6 & -22 & -39 & -49 & -43 & -3 & -109 & -96 \\ 2 & 6 & 7 & 6 & 3 & 1 & 6 & 6 \\ 0 & 0 & 1 & 0 & 0 & 0 & 7 & 7 \\ 0 & 0 & 0 & 1 & 0 & 0 & 6 & 6 \\ 23 & 90 & 179 & 249 & 237 & 11 & 553 & 475 \\ 0 & 1 & 3 & 6 & 7 & 0 & -3 & -6 \\ 0 & -1 & -3 & -6 & -8 & 0 & 0 & 3 \end{bmatrix}$$

(1.5.1)

> E1;

(nu^3).E1,(nu^2).E1;

$$\begin{bmatrix} 1 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 1 \\ -116 \\ 7 \\ 8 \\ 5 \\ 586 \\ 0 \\ -2 \end{bmatrix}$$

(1.5.2)

Also Länge 3.

> E2nn;

(nu^1).E2nn,(nu^0).E2nn;

$$\begin{bmatrix} -16 \\ 451 \\ -48 \\ -32 \\ -16 \\ -2130 \\ 0 \\ 0 \end{bmatrix}$$

$$\begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} -16 \\ 451 \\ -48 \\ -32 \\ -16 \\ -2130 \\ 0 \\ 0 \end{bmatrix}$$

(1.5.3)

Also Länge 1, damit ergibt sich die Partition (3, 1).

Wir betrachten eine weitere Zerlegung:

>  $v, w := \text{Vector}([1, 0, 0, 0, 0, 0, 0, 0]), \text{Vector}([-52/3, 1483/3, -52, -104/3, -52/3, -2342, 0, 0]);$

$$v, w := \begin{bmatrix} 1 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} -\frac{52}{3} \\ \frac{1483}{3} \\ -52 \\ -\frac{104}{3} \\ -\frac{52}{3} \\ -2342 \\ 0 \\ 0 \end{bmatrix}$$

(1.5.4)

>  $(nu^3).v, (nu^2).v;$   
 $(nu^1).w, (nu^0).w;$

$$\begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 1 \\ -116 \\ 7 \\ 8 \\ 5 \\ 586 \\ 0 \\ -2 \end{bmatrix}$$

$$\begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} -\frac{52}{3} \\ \frac{1483}{3} \\ -52 \\ -\frac{104}{3} \\ -\frac{52}{3} \\ -2342 \\ 0 \\ 0 \end{bmatrix}$$

(1.5.5)

Also identische Längen, haben wir auch eine Zerlegung gefunden?

> **Matrix([v,A.v,nu.v,A.nu.v,nu.nu.v,A.nu.nu.v,w,A.w]):**  
**Rank(%);**

8

(1.5.6)

Ja, es war eine. (**DENKANSTOSS:** Warum reicht dies als Verifikation?)

Mithilfe der Partitionen können wir nun endlich unser lang ersehntes Ziel definieren: Die Jordan-Normalform!

**MATH:** Sei  $V$  ein endlich erzeugter  $K$ -Vektorraum und  $\alpha \in \text{End}(V)$  mit charakteristischem Polynom  $\chi_\alpha = p^c$  für ein normiertes irreduzibles Polynom  $p \in K[x]$ . Ist  $a = (a(1), \dots, a(k)) \in \mathbb{N}^k$  die zu  $\alpha$  gehörige Partition von  $c$ , so gibt es eine Basis  $B$  von  $V$  mit

$${}^B_\alpha^B = \text{Diag}(J(p^{a(1)}), \dots, J(p^{a(k)})).$$

Diese Matrix heißt (verallgemeinerte) **Jordan-Normalform**.

**DENKANSTOSSS:** Definiere die Jordan-Normalform für einen beliebigen

Endomorphismus  $\alpha$  durch die Rückführung auf die Haupträume.

Will man lediglich die Jordan-Normalform ohne die Transformationsmatrix, so reicht schon die Partition alleine aus. Hier zwei Möglichkeiten zur Bestimmung der Partition am Beispiel von  $C$ :

```
> C := Matrix([[-585,-474,10119,10546,-14017,-1617,-6895,872,
-3631,-688],[-1382,-1132,23602,24795,-32808,-3775,-16114,
1980,-8495,-1605],[-4029,-3225,69214,72365,-96019,-11064,
-47198,5906,-24873,-4708],[2785,2154,-48563,-50147,67013,
7752,33015,-4316,17372,3303],[-12362,-9644,214162,222347,
-296219,-34207,-145790,18714,-76770,-14569],[-51550,-40704,
890825,926821,-1233228,-142321,-606739,77277,-319553,
-60594],[12360,8726,-220343,-223419,301759,35099,149135,
-20755,78336,14996],[-52629,-45895,881431,941183,-1233773,
-141252,-604243,69661,-319085,-59917],[187520,158113,
-3175625,-3359756,4427554,508348,2171935,-259744,1145853,
215917],[-822332,-701158,13876850,14724709,-19371777,
-2222151,-9497914,1122822,-5012322,-943439]]);
C:= [[-585, -474, 10119, 10546, -14017, -1617, -6895, 872, -3631, -688(1.5.7)
],
[-1382, -1132, 23602, 24795, -32808, -3775, -16114, 1980, -8495,
-1605],
[-4029, -3225, 69214, 72365, -96019, -11064, -47198, 5906,
-24873, -4708],
[2785, 2154, -48563, -50147, 67013, 7752, 33015, -4316, 17372, 3303
],
[-12362, -9644, 214162, 222347, -296219, -34207, -145790, 18714,
-76770, -14569],
[-51550, -40704, 890825, 926821, -1233228, -142321, -606739,
77277, -319553, -60594],
[12360, 8726, -220343, -223419, 301759, 35099, 149135, -20755,
78336, 14996],
[-52629, -45895, 881431, 941183, -1233773, -141252, -604243,
69661, -319085, -59917],
[187520, 158113, -3175625, -3359756, 4427554, 508348, 2171935,
-259744, 1145853, 215917],
[-822332, -701158, 13876850, 14724709, -19371777, -2222151,
-9497914, 1122822, -5012322, -943439]]
> factor(MinimalPolynomial(C,x));
factor(CharacteristicPolynomial(C,x));
(x-2)3
(x-2)10 (1.5.8)
```

> **CC:=C-2:**

Erste Möglichkeit über **Bilder**:

> **Rank(IdentityMatrix(10)),Rank(CC),Rank(CC^2),Rank(CC^3);**  
10, 5, 2, 0 (1.5.9)

> **(10-5,5-2,2-0);**  
5, 3, 2 (1.5.10)

ist die konjugierte Partition.

Zweite Möglichkeit über **Kerne**:

> **nops(NullSpace(IdentityMatrix(10))),nops(NullSpace(CC)),nops**  
**(NullSpace(CC^2)),nops(NullSpace(CC^3));**  
0, 5, 8, 10 (1.5.11)

> **(5-0,8-5,10-8);**  
5, 3, 2 (1.5.12)

ist wieder die konjugierte Partition.

Dass dieses Verfahren funktioniert liegt an der Beweisidee der Eindeutigkeit der zu  $\alpha$ -gehörigen Partition: Definiere  $V_0 := V$  und  $V_i := \text{Bild}(v^i)$  für  $i > 0$ . Dann liefert jede Zerlegung von  $V$  auch Zerlegungen der  $V_i$  und wir erkennen: Die zu  $\alpha$ -gehörige Partition ist konjugiert zu folgender:  $(\text{Dim}_{K[X]/pK[X]}(V_0/V_1), \text{Dim}_{K[X]/pK[X]}(V_1/V_2), \dots, \text{Dim}_{K[X]/pK[X]}(V_{m-1}/V_m))$ , wobei  $m$  minimal ist mit  $V_m = \{0\}$ .

Insbesondere muss man aufpassen, falls  $p$  nicht Grad 1 hat:

**BEISPIEL:**

> **X:=DiagonalMatrix([JBlock(x^2+1,x,2),JBlock(x^2+1,x,1)]);**

$$X := \begin{bmatrix} 0 & -1 & 0 & 0 & 0 & 0 \\ 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & -1 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & -1 \\ 0 & 0 & 0 & 0 & 1 & 0 \end{bmatrix} \quad (1.5.13)$$

> **XX:=X^2+1;**

(1.5.14)

$$XX := \begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix} \quad (1.5.14)$$

```
> nops(NullSpace(IdentityMatrix(10))),nops(NullSpace(XX)),nops
  (NullSpace(XX^2));
                                0, 4, 6                                (1.5.15)
```

```
> (4-0,6-4);
                                4, 2                                (1.5.16)
```

**DENKANSTOSS:** Warum ist die zugehörige Partition nicht (4, 2) sondern (2, 1) ?

### ÜBUNG [07]:

- 1.) Schreibe mit Hilfe der eben ausgerechneten Partition die Jordan-Normalform von  $C$  hin, ohne die zugehörige Basistransformation auszurechnen.
- 2.) Schreibe alle möglichen Partitionen eines Endomorphismus mit Minimalpolynom  $p^3$  und charakteristischem Polynom  $p^{10}$  hin, wobei  $p$  ein beliebiges Polynom von Grad 1 ist. (Hinweis: Du darfst dir Hilfe in Maples **combinat**-Paket suchen.) Schreibe für einige von ihnen die zugehörige Jordan-Form für den Fall  $p = x - a$  hin.

## ▼ Jordan-Normalform

```
> interface(rtablesize=20):#damit Maple die nächste Matrix
  auch anzeigen kann
```

**MATH:** Mit Hilfe der vorgestellten Verfahren lässt sich die Jordan-Normalform einer beliebigen Matrix herstellen. Sie ist eindeutig bis auf die Reihenfolge der Haupträume, die Basis ist es allerdings nicht.

### ÜBUNG [08]:

Bestimme eine Basis, die die folgende Matrix  $E$  über  $\mathbb{Q}$  auf Jordan-Normalform transformiert, sowie die zugehörige Jordan-Normalform.

```
E:=Matrix([[275021318662427290940,-921450713945492110586,
2453606792486202458837,-3837810574795276766056,
4001130293392669817196,-5087428337352354046535,
1441145746614411874471,2082203903468764335393,
```

-4941456758923748098857,3494002585770741408845,  
-2763658307219555998694,3136403345001268019758,  
-4614177252374464554244,-5573936352142738880160,  
-1740530619726901686759],[54735511096680626937,  
-183389695109836703838,488323677852276332288,  
-763811781302332759419,796316127921973910781,  
-1012514199142919431122,286820852250257982112,  
414406037387081298268,-983462527840717281338,  
695386154918515629660,-550030996425862264857,  
624215755087340270021,-918326446210505045987,  
-1109340383279099019042,-346405266011800253538],  
[-5275394552374491653,17675051884236843351,  
-47064511106640583012,73615984018209735040,  
-76748744627366552161,97585859405345025315,  
-27643720340818731740,-39940347834542928648,  
94785866759724034134,-67021139657838133863,  
53011846679514659677,-60161754735013038917,  
88508067880891965833,106917942254027654033,  
33386450891171265778],[2722077124931574669,  
-9120238105110811998,24285051631220596041,  
-37985478457739415141,39601967216525079856,  
-50353813912271975978,14264021020736461810,  
20609019121287091957,-48909031753335464537,  
34582571850920528406,-27353846951928033916,  
31043163640947221418,-45669719024527696035,  
-55169121847662584254,-17227241176226856157],  
[-528799482172912163,1771726871036938419,  
-4717692459744254515,7379181564904368022,  
-7693205885066025699,9781894303567892888,  
-2770974731165500059,-4003574527559439064,  
9501226261321512649,-6718121951608589991,  
5313846536945706452,-6030545096606906002,  
8871946922429627232,10717331554561191957,  
3346619436246007149],[-56716525441546926,  
190027024504135134,-505997326848606973,791456030259155650,  
-825136790063432418,1049159607447197330,  
-297201612588979076,-429404423045922801,  
1019056483871269229,-720553910193089963,569937986827571811,  
-646807676498875623,951562965368855501,1149490172119634449,  
358942534548367988],[6829498508865867,-22882030773060237,  
60929472711965092,-95302872247713576,99358527933107882,  
-126334148977606832,35787417409820951,51706567778311656,  
-122709292977182574,86765220840223845,-68628862591427668,  
77885096588431182,-114582087010640512,-138415415178112556,  
-43221926685093820],[1685737437363180,-5648012934925324,  
15039331666203395,-23523779883308889,24524844692503489,  
-31183285899000504,8833472799747367,12762825403382269,  
-30288556153100771,21416415984421111,-16939785959843123,



19224518895633400,-28282503242889455,-34165326629829128,  
-10668546136044475],[ -570726205014827,1912201103665996,  
-5091742342330493,7964251918987695,-8303174165646742,  
10557467626088858,-2990675947740913,-4321004414036086,  
10254546363840064,-7250779124286824,5735163460459944,  
-6508686625443272,9575373594033308,11567072533601344,  
3611961574981558],[283081090861199,-948454739968805,  
2525512170274536,-3950281415701150,4118387380455265,  
-5236520464747078,1483379950427389,2143224951641034,  
-5086271044617423,3596397792967601,-2844650051808365,  
3228318752178110,-4749400287903652,-5737286078890674,  
-1791538593833784],[ -21007977229181,70386600247219,  
-187422981892585,293157772488464,-305633230561086,  
388611978106093,-110084400645203,-159052732361623,  
377461687614661,-266895406942102,211106800999043,  
-239579572866461,352461878674362,425774730969122,  
132953429952770],[ -11562714168080,38740528469704,  
-103156926741552,161352970464701,-168219417157065,  
213890617648291,-60590053248774,-87542044712814,  
207753538360141,-146898260097180,116192414534242,  
-131863724492416,193993734553842,234344861495517,  
73177083705453],[3641290229729,-12200034157986,  
32485825042999,-50812723237425,52975081045115,  
-67357698629971,19080811452471,27568440028181,  
-65425030701240,46260695498690,-36590916083979,  
41526071186902,-61091840549379,-73799078845857,  
-23044675849098],[337363329312,-1130325758619,  
3009791969445,-4707767961695,4908109099879,-6240649888045,  
1767825597898,2554199233000,-6061589377696,4286025355034,  
-3390126162356,3847365286742,-5660121940604,-6837439854769,  
-2135075227981],[33396666821,-111894534700,297948860720,  
-466036893853,485869299046,-617781741388,175002667344,  
252848289565,-600055973118,424287254545,-335599349746,  
380862901982,-560313437195,-676859874525,-211357873947]]):

Dimension(E);

15, 15

(1.6.1)